```
> restart;
> with(inttrans):
  with(plots):
```

```
> # Бесконечная синусоида частоты 1 Гц
```

```
> signal:=sin(2*Pi*t);
```

$$signal := \sin(2\,\pi\,t) \tag{1}$$

```
> signal_plot:=plot(signal, t=-6..6, color=blue, thickness=2):
  display(signal_plot);
```




```
> # Ограничивающее окно длительностью 3 секунды
```

```
> window:=piecewise(t >= 0 and t <= 3, 1, 0);
```

$$window := \begin{cases} 1 & 0 \le t \text{ and } t \le 3 \\ 0 & otherwise \end{cases} \tag{2}$$

```
> window_plot:=plot(window, t=-6..6, color=yellow, thickness=4):
> display(window_plot, signal_plot);
```




```
> # Перемножаем сигнал на окно и получаем "кусочек синуса"
```

```
> signal_windowed:=simplify(signal*window);
```

(3)

$$signal\_windowed := \begin{cases} 0 & t < 0 \\ \sin(2\,\pi\,t) & t \le 3 \\ 0 & 3 < t \end{cases} \quad (3)$$

```
> plot(signal_windowed, t=-6..6, color=blue, thickness=2);
```




```
> # Частота дискретизации в герцах и в рад/с
> sample_freq_hz:=3;
  sample_freq:=2*Pi * sample_freq_hz;
```

$$sample\_freq\_hz := 3$$

$$sample\_freq := 6\,\pi \quad (4)$$

```
> # Формируем идеальный НЧ-фильтр с частотой среза sample_freq/2
> filter_freq_response:=piecewise(abs(w) < sample_freq/2, 1, 0);
```

$$filter\_freq\_response := \begin{cases} 1 & |w| < 3\,\pi \\ 0 & otherwise \end{cases} \quad (5)$$

```
> plot(filter_freq_response, w=-20..20, color=green, thickness=4)
  ;
```




```
> # Получаем импульсный отклик, применив обратное преобразование Фурье.
> filter_impulse_response:=invfourier(filter_freq_response, w, t)
  ;
```

(6)

$$filter\_impulse\_response := \frac{\sin(3\,\pi\,t)}{t\,\pi} \quad \textbf{(6)}$$

```
> plot(filter_impulse_response, color=blue, thickness=2);
```




```
> # Применение фильтра - это свертка сигнала с его импульсным откликом
> convolve:=(f,g)->int(f(tau)*g(t-tau), tau=-infinity..infinity);
```

$$convolve := (f, g) \to \int_{-\infty}^{\infty} f(\tau)\, g(t-\tau)\, d\tau \quad \textbf{(7)}$$

```
> # Результат удается выразить аналитически через интегральные синусы и косинусы
> signal_filtered:=convolve(x->subs(t=x,signal_windowed), x->subs
  (t=x,filter_impulse_response));
```

$$signal\_filtered := \frac{1}{2}\,\frac{1}{\pi}\left(2\,\mathrm{Si}(\pi t)\sin(\pi t)\cos(\pi t) - 2\,\mathrm{Ci}(-\pi t)\cos(\pi t)^2 + \mathrm{Ci}(-\pi t)\right. \quad \textbf{(8)}$$

$$+ 2\,\mathrm{Si}(5\,\pi t)\sin(\pi t)\cos(\pi t) + 2\,\mathrm{Ci}(-5\,\pi t)\cos(\pi t)^2 - \mathrm{Ci}(-5\,\pi t) + 2\,\mathrm{Si}(-\pi t$$

$$+ 3\,\pi)\sin(\pi t)\cos(\pi t) + 2\,\mathrm{Ci}(-\pi t + 3\,\pi)\cos(\pi t)^2 - \mathrm{Ci}(-\pi t + 3\,\pi) + 2\,\mathrm{Si}($$

$$-5\,\pi t + 15\,\pi)\sin(\pi t)\cos(\pi t) - 2\,\mathrm{Ci}(-5\,\pi t + 15\,\pi)\cos(\pi t)^2 + \mathrm{Ci}(-5\,\pi t$$

$$\left. + 15\,\pi)\right)$$

```
> display(
    window_plot,
    plot(signal_filtered, t=-6..6, color=blue, thickness=2)
  );
```


1
0.5
-0.5
-1
-6
-4
-2
0
2
4
6
t